

COMP30080 Assignment 1
Sample Solution
Pavel Gladyshev, <student number>

Q1.(a)

Reads the first elements from arrays D1 and D2

Adds the value read from array D2 from the value read from array D1

Stores the result in the first element of D3

In other words, it performs: $D3[0] = D1[0] + D2[0]$

Q1.(b)

```
.data                                # data goes in data segment
D1:    .word    1,2,4                # data stored in words
D2:    .word    5,6,8
D3:    .word    0,0,0

.text                                # code goes in text segment
.globl main                          # must be global symbol
main:  la        $t0, D1              # load address pseudo-instruction
      la        $t1, D2
      la        $t2, D3
      # Index 0
      lw        $t3, 0($t0)
      lw        $t4, 0($t1)
      add       $t3, $t3, $t4
      sw        $t3, 0($t2)
      # Index 1
      lw        $t3, 4($t0)
      lw        $t4, 4($t1)
      add       $t3, $t3, $t4
      sw        $t3, 4($t2)
      # Index 2
      lw        $t3, 8($t0)
      lw        $t4, 8($t1)
      add       $t3, $t3, $t4
      sw        $t3, 8($t2)
      #
      li        $v0, 10              # system call for exit
      syscall                          # Exit!
```

Q1.(c)

```
.data                                # data goes in data segment
D1:    .word    1,2,4                # data stored in words
D2:    .word    5,6,8
D3:    .word    0,0,0

.text                                # code goes in text segment
.globl    main                       # must be global symbol
main:    la      $t0, D1              # $t0 is the address of D1[0]
        la      $t1, D2              # $t1 is the address of D2[0]
        la      $t2, D3              # $t2 is the address of D3[0]
        li      $t5, 3.              # $t5 is number of elements per array
loop:    lw      $t3, 0($t0)           # $t3 is the next element of D1
        lw      $t4, 0($t1)           # $t4 is the next element of D2
        add     $t3, $t3, $t4
        sw      $t3, 0($t2)
        addiu   $t0, $t0, 4           # increment D1 pointer
        addiu   $t1, $t1, 4           # increment D2 pointer
        addiu   $t2, $t2, 4           # increment D3 pointer
        addiu   $t5, $t5, -1          # decrement loop counter
        bnez    $t5, loop             # if loop counter not zero, keep looping
        #
        li      $v0, 10               # system call for exit
        syscall                       # Exit!
```

Q2.

```
.data
N:    .word 5          # the number of elements in x[]
x:    .word 1,2,3,4,5  # array of integers x[]
.text
.globl main

main:
    lw     $t1, N       # $t1 = value of variable N
    la     $t2, x       # $t2 = address of the first element in x[]
    sll    $t3, $t1, 2   # $t3 = $t1*4
    addu   $t3, $t3, $t2
    addiu  $t3, $t3, -4  # $t3 points to the last element of x[]
    srl    $t1, $t1, 1   # $t1 = number of elements in x[] / 2

loop:
    beqz   $t1, exit     # while ($t1 != 0)
    # swap elements of the array pointed to by $t2 and $t3
    lw     $t4, 0($t2)
    lw     $t5, 0($t3)
    sw     $t5, 0($t2)
    sw     $t4, 0($t3)
    # move array pointers
    addiu  $t2, $t2, 4
    addiu  $t3, $t3, -4
    # decrement loop counter
    addiu  $t1, $t1, -1
    j      loop

exit:
    li     $v0, 10
    syscall
```